

Mathematics

HP COMPUTER CURRICULUM

Linear Equations & Systems

TEACHERS ADVISOR

HEWLETT hp PACKARD

HEWLETT hp PACKARD

Hewlett-Packard
Computer Curriculum Series

mathematics
TEACHER'S ADVISOR

linear equations & systems

by E.A. Herber
Golden High School
Golden, Colorado

edited by Jean Danver and Christine Doerr
Hewlett-Packard

Hewlett-Packard Company
11000 Wolfe Road
Cupertino, California 95014

HP 5951-4448

June 1973

The Hewlett-Packard Computer Curriculum Series represents the published results of a Curriculum Development project sponsored by the Data Systems Division of Hewlett-Packard Company. This project is under the directorship of Jean H. Danver.

This material is designed to be used with any Hewlett-Packard system with the BASIC program language such as the 9830A Educational BASIC, and the 2000 and 3000 series systems.

The Hewlett-Packard Company makes no warranty, expressed or implied, and assumes no responsibility in connection with the operation of the program material attached hereto.

TABLE OF CONTENTS

1 INTRODUCTION

3 LINEAR EQUATIONS
3 Exercise 1 — Converting the General Form to Slope-Intercept Form
5 Exercise 2 — Finding the General Form from Characteristics of the Line
7 Exercise 3 — Determining Collinearity
10 Exercise 4 — Finding the Length and Midpoint of a Given Line
11 Exercise 5 — Equation for the Perpendicular Bisector of a Line
12 Exercise 6 — The Intersection of Two Lines
14 Exercise 7 — Finding the Center and Radius of a Circle Determined by Three Points

21 LINEAR SYSTEMS
21 Exercise 8 — An Iterative Method
22 Exercise 9 — Applying the Gaussian Algorithm
29 Exercise 10 — Solving Linear Systems by Use of Matrices

31 LINEAR PROGRAMMING
31 Exercise 11 — A Manufacturing Problem

INTRODUCTION

This Mathematics Set of the Hewlett-Packard Computer Curriculum Series consists of a set of a Student Book and a corresponding Teacher's Advisor. It was designed to help meet the need for computer-oriented problems in mathematics, providing students an opportunity to use a computer as a problem solving tool to model mathematical concepts.

The materials are designed for flexible use as desired by the individual instructor. The material and exercises in this unit are intended as an "enrichment" experience in the field of linear equations and systems. The unit begins with a discussion of the various forms for a linear equation. The student is taught to derive the equation for a line when given a set of data such as two points on the line, one point and the slope, etc. Other exercises involve applications of this technique. The second section deals with solving linear systems. Two methods are discussed: an Iterative Method for solving low order systems, and the Gaussian Algorithm for higher order systems. The last section is a brief introduction to linear programming. The unit is obviously not a complete study of linear equations and systems and should not compete with your text. The material within each section is sequential, but the sections can be studied independently and in any order. Thus, the material can be used to supplement and enrich your curriculum in any fashion you choose.

The mathematical concepts needed for each exercise are briefly reviewed, but you may want your students to study these in greater detail before attempting the exercises, especially if their background in linear equations is limited. A list of possible references is included at the end of each of the three major sections. These references will also provide additional problems for your students.

The degree of difficulty of the material is dependent upon the amount of assistance which you choose to provide. With no assistance, the better student should be challenged. However, given a good deal of assistance, any second year algebra student should be able to work out the exercises with no great difficulty. The level of the material is determined by the assumption that students taking second year algebra will be quite capable as a group.

The Student Book provides text material and programming exercises for the students. There is a problem analysis, including a suggested approach and a macro flowchart, for each exercise. The Teacher's Advisor contains an example of a program to solve each exercise, micro flowcharts, and a brief discussion of the important elements of the exercise. The micro flowcharts should be given to the students only after they have made an attempt to solve the problem on their own.

For best results, you should study all the solutions until you are certain you have a complete grasp of the general methods. This should be done before assigning the material to the class.

You will undoubtedly think of different programming methods or techniques as you study the example programs. Encourage the students to do the same. There are no *approved* solutions. All solutions are acceptable if they produce the correct results. At this level, there is no need for emphasis on the efficiency of a student's program. The important question is, does it work?

LINEAR EQUATIONS

Exercise 1 — Converting the General Form to Slope-Intercept Form

This program models the definition of slope and y-intercept, and shows the relation between the general form and the y-intercept form of the equation of a line.

The program will not require any sophisticated programming techniques, but it does model those basic concepts involved in the study of linear equations. Make sure the student's program makes provision for consideration of horizontal and vertical lines.

This program can be used to introduce "string variables", if you have them available on your system, by assigning the string "Undefined" to m if the slope is undefined.

Micro Flowchart Exercise 1



Example Program
Exercise 1

```
10  DIM M$[9]
20  DIM K$[4]
30  REM--THIS PROGRAM TAKES  THE COEFFICIENTS OF A LINEAR EQUATION
40  REM--IN GENERAL FORM, AX+BY+C = 0, COMPUTES THE SLOPE AND
50  REM--Y-INTERCEPT OF THE LINE AND PRINTS THE SLOPE-INTERCEPT
60  REM--FORM OF THE EQUATION.  ON THE DATA LINE ENTER THE
70  REM--COEFFICIENTS A,B AND C.
80  DATA 3,-2,7,.23,.1,-.6,8,-2,-7,3,0,4,0,2,-5
90  READ A,B,C
100  IF B=0 THEN 170
110  LET K=-C/B
120  LET M=-A/B
130  PRINT "SLOPE:";TAB(15);M
140  PRINT "Y-INT:";TAB(15);K
150  PRINT "EQUATION:";TAB(15);"Y=";M;"X+";K
160  GOTO 220
170  LET M$="UNDEFINED"
180  LET K$="NONE"
190  PRINT "SLOPE:";TAB(15);M$
200  PRINT "Y-INT:";TAB(15);K$
210  PRINT " EQUATION:";TAB(15);"X=";-C/A
220  GOTO 90
230  END
```

```
RUN

SLOPE:           1.5
Y-INT:           3.5
EQUATION:       Y= 1.5          X+ 3.5
SLOPE:          -2.3
Y-INT:           6.
EQUATION:       Y=-2.3          X+ 6.
SLOPE:           4
Y-INT:          -3.5
EQUATION:       Y= 4     X+-3.5
SLOPE:          UNDEFINED
Y-INT:          NONE
 EQUATION:      X=-1.33333
SLOPE:           0
Y-INT:           2.5
EQUATION:       Y= 0     X+ 2.5

OUT OF DATA  IN LINE 90
```

Exercise 2 — Finding the General Form from Characteristics of the Line

This exercise makes the student aware of the relationship of the coefficients of the general form and the slope-intercept form of linear equations. Make sure your students understand the development of the different equations, since this is the real value of the exercise.

Micro Flowchart
Exercise 2.



Example Program
Exercise 2.

```
10  REM--A PROGRAM TO PRINT THE GENERAL FORM OF AN EQUATION OF A
20  REM-- LINE GIVEN:  CASE (1)TWO POINTS ON THE LINE,
30  REM-- (2)A POINT ON THE LINE AND THE SLOPE OF THE LINE,
40  REM (3)A POINT ON THE LINE AND THE Y-INTERCEPT OF THE LINE.
50  REM--(4) THE SLOPE OF OF THE LINE AND THE Y-INTERCEPT OF THE LINE.
60  REM--ENTER ON THE DATA LINE EACH SET OF DATA TO BE
70  REM-- CONSIDERED PRECEDED BY ITS CASE NUMBER AS NUMBERED ABOVE.
80  DATA 1,2,-1,-3.2,7,1,0,8,-6,0,1,.8,-2,16,-2,1,-1.5,2,-1.5,-8
90  DATA 2,-7.5,-3,-3,3,3.27,-6,-5,4,.875,7.3,4,2.5,0,4,0,2
100  READ K
110  PRINT
120  IF K=1 THEN 230
130  IF K=2 THEN 280
140  IF K=3 THEN 330
150  READ M,B1
160  A=-M
170  B=1
180  C=-B1
190  IF A=0 THEN 370
200  IF B=0 THEN 390
210  PRINT A;"X+";B;"Y+";C;"=0"
220  GOTO 100
230  READ X1,Y1,X2,Y2
240  A=Y1-Y2
250  B=X2-X1
260  C=X1*Y2-Y1*X2
270  GOTO 190
280  READ X1,Y1,M
290  A=M
300  B=-1
310  C=-M*X1+Y1
320  GOTO 190
330  READ X1,Y1,B
340  X2=0
350  Y2=B
360  GOTO 240
370  PRINT B;"Y+";C;"=0"
380  GOTO 100
390  PRINT A;"X+";C;"=0"
400  GOTO 100
410  END
```

```
RUN

-8     X+-5.2          Y+ 10.8         =0
 8     X+-6     Y+ 48    =0
 15.2         Y+ 30.4         =0
 10    X+ 15    =0
-3     X+-1     Y+-25.5         =0
-1     X+-3.27          Y+-16.35        =0
-.875          X+ 1     Y+-7.3          =0
-2.5           X+ 1     Y+ 0     =0
 1     Y+-2     =0
OUT OF DATA  IN LINE 100
```

Exercise 3 — Determining Collinearity

Exercises 3-6 are really sub-programs of Exercise 7. The students should fully understand these four exercises before attempting Exercise 7.

Micro Flowchart Exercise 3.



Example Program
Exercise 3.

```
10  REM--A PROGRAM TO DETERMINE IF A SET OF POINTS IS COLLINEAR.
11  REM-- ENTER ON THE DATA LINE N, THE NUMBER OF POINTS IN THE SET
12  REM--FOLLOWED BY THE COORDINATES OF THE QOINTS.
20  DATA 4,-3,-2,0,0,6,4,3,2,3,-1,5,3,5,4.5,5,4,2,-5,-5,-10,2,-5,3,6
21  DATA 4,1,2,-5,-10,2,-5,3,6,4,2,0,2,-3.5,2,8,2,-7
30  READ N
40  FOR I=1 TO N
50  READ X[I],Y[I]
60  NEXT I
80  I=1
90  J=2
100  IF X[I]=X[J] THEN 190
110  I=1
120  J=2
121  IF X[I]=X[J] THEN 150
122  IF X[I]=X[J+1] THEN 150
130  M=(Y[I]-Y[J])/(X[I]-X[J])
140  IF M=(Y[I]-Y[J+1])/(X[I]-X[J+1]) THEN 220
150  PRINT "POINTS ARE NOT COLLINEAR"
180  GOTO 30
190  IF J=N THEN 250
200  J=J+1
210  GOTO 100
220  IF J=N-1 THEN 270
230  J=J+1
240  GOTO 140
250  PRINT "POINTS DETERMINE A VERTICAL LINE"
260  GOTO 30
270  IF M=0 THEN 300
280  PRINT "POINTS DETERMINE A LINE WITH THE SLOPE ";M
290  GOTO 30
300  PRINT "POINTS DETERMINE A HORIZONTAL LINE"
310  GOTO 30
320  END


RUN


POINTS DETERMINE A LINE WITH THE SLOPE  .666667
POINTS DETERMINE A HORIZONTAL LINE
POINTS ARE NOT COLLINEAR
POINTS ARE NOT COLLINEAR
POINTS DETERMINE A VERTICAL LINE

OUT OF DATA  IN LINE 30
```

Exercise 4 — Finding the Length and Midpoint of a Given Line

This exercise is relatively simple but should still be completed in preparation for Exercise 7.

The flowchart in the Student Book is detailed enough to produce the following program.

Example Program
Exercise 4.

```
10  REM--A PROGRAM TO DETERMINE THE LENGTH AND MID-POINT OF THE
11  REM--SEGMENT DETERMINED BY TWO POINTS ON THE CARTESIAN PLANE.
12  REM--ENTER THE COORDINATES OF THE POINTS ON THE DATA LINE.
20  DATA -3,5,2,-7,-7,-1,8,5,-5,7,3.5,7,-4,-6.25,-4,4.8
30  READ X1,Y1,X2,Y2
40  D=SQR((X2-X1)↑2+(Y2-Y1)↑2)
50  M=(X2+X1)/2
60  N=(Y2+Y1)/2
70  PRINT "LENGTH OF SEGMENT P1,P2 IS";D
80  PRINT "MID-POINT, (M,N) ,OF SEGMENT P1,P2 IS (";M;",";N;")"
85  GOTO 30
90  END


RUN


LENGTH OF SEGMENT P1,P2 IS 13
MID-POINT, (M,N) ,OF SEGMENT P1,P2 IS (-.5          ,-1     )
LENGTH OF SEGMENT P1,P2 IS 16.1555
MID-POINT, (M,N) ,OF SEGMENT P1,P2 IS ( .5          , 2     )
LENGTH OF SEGMENT P1,P2 IS 8.5
MID-POINT, (M,N) ,OF SEGMENT P1,P2 IS (-.75         , 7     )
LENGTH OF SEGMENT P1,P2 IS 11.05
MID-POINT, (M,N) ,OF SEGMENT P1,P2 IS (-4     ,-.725        )

OUT OF DATA  IN LINE 30
```

Exercise 5 — Equation for the Perpendicular Bisector of a Line

The student will not encounter any new concepts in this exercise. Concepts of previous exercises are applied.

Micro Flowchart Exercise 5.



Example Program
Exercise 5.

```
10   REM--A PROGRAM TO FIND THE EQUATION OF THE PERPENDICULAR BISECTOR
20   REM--OF A LINE SEGMENT DETERMINED BY TWO POINTS IN THE CARTESIAN
30   REM--PLANE.  ON THE DATA LINE ENTER THE COORDINATES OF
40   REM--THE TWO POINTS.
50   DATA -7,2.5,-6.7,0,7.17,-11.5,-3.1,-3
60   READ X1,Y1,X2,Y2
70   X=(X1+X2)/2
80   Y=(Y1+Y2)/2
90   IF Y1=Y2 THEN 150
100   IF X1=X2 THEN 170
110   M=-(X2-X1)/(Y2-Y1)
120   B=Y-M*X
130   PRINT "EQUATION OF LINE L IS Y=";M;"*X+";B
140   GOTO 180
150   PRINT "THE EQUATION OF LINE L IS X=";X
160   GOTO 180
170   PRINT "THE EQUATION OF LINE L IS Y=";Y
180   GOTO 60
190   END

RUN

EQUATION OF LINE L IS Y= 1.5          *X+-.5
THE EQUATION OF LINE L IS Y= 8.5
THE EQUATION OF LINE L IS X=-5.25

OUT OF DATA  IN LINE 60
```

Exercise 6 — The Intersection of Two Lines

This exercise deals only with the solution of a second order linear system. Once again this is a sub-program of Exercise 7. Higher order systems will be considered in the section on Linear Systems.

It is very important that your students fully understand the development of the expressions which represent the coordinates of the point of intersection. Encourage them to consult reference works. Not much value comes from just having the computer evaluate the expressions

$$\frac{B_1C_2 - B_2C_1}{A_1B_2 - A_2B_1} \text{ and } \frac{A_2C_1 - A_1C_2}{A_1B_2 - A_2B_1}$$

The value of the exercise comes from understanding the derivation of the expressions and the conditions under which unique solutions occur.

Micro Flowchart
Exercise 6.



Example Program
Exercise 6.

```
10  REM--A PROGRAM TO DETERMINE THE COORDINATES OF
11  REM--THE POINT OF INTERSECTION OF TWO
20  REM--LINES, GIVEN THEIR EQUATIONS IN GENERAL FORM.
30  REM--ON THE DATA LINES ENTER THE COEFFICIENTS OF THE TWO
40  REM--LINEAR EQUATIONS.
50  DATA 2,-1,3,1,1,-9,.25,1,-2,1,4,2,2,-2,-99,.5,.6,-14,1,2,-3,4,8,-12
60  READ A1,B1,C1,A2,B2,C2
70  IF A1*B2-A2*B1=0 THEN 110
80  PRINT "THE POINT OF INTERSECTION IS (";(B1*C2-B2*C1)/(A1*B2-A2*B1);
90  PRINT ",";(A2*C1-A1*C2)/(A1*B2-A2*B1);")"
100  GOTO 60
110  IF B1*C2-B2*C1 <> 0 THEN 150
120  IF A2*C1-A1*C2 <> 0 THEN 150
130  PRINT "LINES ARE C0-INCIDENT"
140  GOTO 60
150  PRINT "LINES ARE PARALLEL"
160  GOTO 60
170  END
```

```
RUN

THE POINT OF INTERSECTION IS ( 2      , 7      )
LINES ARE PARALLEL
THE POINT OF INTERSECTION IS ( 39.7273      ,-9.77273      )
LINES ARE C0-INCIDENT

OUT OF DATA  IN LINE 60
```

Exercise 7 — Finding the Center and Radius of a Circle Determined by Three Points

This exercise requires combining several of the previous exercises. It is expected that the first programs students write for this exercise might be rather lengthy. The flowchart and program given here are moderately detailed and lengthy, respectively. Not counting the REM lines, the example program has 53 lines.

After your students have their programs running, challenge them to reduce the code to as few lines as possible. The alternate program included here has 26 lines, but it has been written in as few as 10 lines by some students.

Micro Flowchart
Exercise 7.




3
1
2
$x_1 = x_3$
YES
$A_2 \leftarrow 0$
$B_2 \leftarrow 1$
$C_2 \leftarrow -\frac{(y_1+y_3)}{2}$
NO
$y_1 = y_3$
YES
$A_2 \leftarrow 1$
$B_2 \leftarrow 0$
$C_2 \leftarrow -\frac{(x_1+x_3)}{2}$
NO
$S_2 \leftarrow \frac{x_1+x_3}{2}$
$T_2 \leftarrow \frac{y_1+y_3}{2}$
$M_2 \leftarrow \frac{(x_1-x_3)}{y_1-y_3}$
$A_2 \leftarrow M_2$
$B_2 \leftarrow -1$
$C_2 \leftarrow T_2-M_2S_2$
$h \leftarrow \frac{B_1C_2-B_2C_1}{A_1B_2-A_2B_1}$
$k \leftarrow \frac{A_2C_1-A_1C_2}{A_1B_2-A_2B_1}$
$r \leftarrow SQR\left((x_1-h)^2 + (y_1-k)^2\right)$
OUTPUT
CENTER: (h,k)
RADIUS: r
STOP

Example Program
Exercise 7.

```
10  REM--A PROGRAM TO FIND THE RADIUS AND CENTER OF A CIRCLE
20  REM--DETERMINED BY THREE NON-COLLENEAR POINTS IN THE CARTESIAN
30  REM--PLANE.  ON THE DATA LINE ENTER THE COORDINATES OF THE TRIPLETS
40  REM--OF POINTS TO BE CONSIDERED.
50  DATA -2,3,7,6,5,-4,6,1,-2,1,4,3,7,2,-3,-5,7,-1,3,1,3,-1,3,6
60  READ X1,Y1,X2,Y2,X3,Y3
70  IF X1=X2 AND X2=X3 THEN 300
80  IF Y1=Y2 AND Y2=Y3 THEN 330
90  IF (X1-X3)*(Y1-Y2)=(Y1-Y3)*(X1-X2) THEN 360
100  IF X1=X2 THEN 390
110  IF Y1=Y2 THEN 430
120  S1=(X1+X2)/2
130  T1=(Y1+Y2)/2
140  M1=-(X1-X2)/(Y1-Y2)
150  A1=M1
160  B1=-1
170  C1=T1-M1*S1
180  IF X1=X3 THEN 470
190  IF Y1=Y3 THEN 510
200  S2=(X1+X3)/2
210  T2=(Y1+Y3)/2
220  M2=-(X1-X3)/(Y1-Y3)
230  A2=M2
240  B2=-1
250  C2=T2-M2*S2
260  H=(B1*C2-B2*C1)/(A1*B2-A2*B1)
270  K=(A2*C1-A1*C2)/(A1*B2-A2*B1)
280  R=SQR((X1-H)↑2+(Y1-K)↑2)
290  GOTO 540
300  PRINT "POINTS LIE IN A VERTICAL LINE."
310  PRINT
320  GOTO 60
330  PRINT "POINTS LIE IN A HORIZONTAL LINE."
340  PRINT
350  GOTO 60
360  PRINT "POINTS LIE ON AN OBLIQUE LINE."
370  PRINT
380  GOTO 60
390  A1=0
400  B1=1
410  C1=-(Y1+Y2)/2
420  GOTO 190
430  A1=1
440  B1=0
450  C1=-(X1+X2)/2
460  GOTO 180
470  A2=0
480  B2=1
490  C2=-(Y1+Y3)/2
```

```
500  GOTO 260
510  A2=1
520  B2=0
530  C2=-(X1+X3)/2
535  GOTO 260
540  PRINT "CENTER: (";H;",";K;")"; RADIUS:"R
550  PRINT
560  GOTO 60
570  END


RUN

CENTER: ( 3.5          , 1.5          ); RADIUS: 5.70088

CENTER: ( 2      ,-1      ); RADIUS: 4.47214

CENTER: ( .6           , .5           ); RADIUS: 6.57343

POINTS LIE IN A VERTICAL LINE.

OUT OF DATA  IN LINE 60
```

Alternate Flowchart
Exercise 7.



Alternate Program
Exercise 7.

```
10  REM--A PROGRAM TO FIND THE RADIUS AND CENTER OF A CIRCLE
20  REM--DETERMINED BY THREE NON-COLLENEAR POINTS IN THE CARTESIAN
30  REM--PLANE.  ON THE DATA LINE ENTER THE COORDINATES OF THE TRIPLETS
40  REM--OF POINTS TO BE CONSIDERED.
50  DATA -2,3,7,6,5,-4,6,1,-2,1,4,3,7,2,-3,-5,7,-1,3,1,3,-1,3,6
60  READ X1,Y1,X2,Y2,X3,Y3
70  IF (X1-X3)*(Y1-Y2)=(Y1-Y3)*(X1-X2) THEN 200
80  IF Y1=Y2 THEN 220
90  A1=-(X1-X2)/(Y1-Y2)
100  B1=-1
110  C1=(Y1+Y2)/2-A1*(X1+X2)/2
120  IF Y1=Y3 THEN 260
130  A2=-(X1-X3)/(Y1-Y3)
140  B2=-1
150  C2=(Y1+Y3)/2-A2*(X1+X3)/2
160  H=(B1*C2-B2*C1)/(A1*B2-A2*B1)
170  K=(A2*C1-A1*C2)/(A1*B2-A2*B1)
180  PRINT "CENTER: (";H;",";K;")"; RADIUS: ";SQR((X1-H)↑2+(Y1-K)↑2)
190  GOTO 60
200  PRINT "POINTS ARE COLLINEAR"
210  GOTO 60
220  A1=1
230  B1=0
240  C1=-(X1+X2)/2
250  GOTO 120
260  A2=1
270  B2=0
280  C2=-(X1+X3)/2
290  GOTO 160
300  END
```

```
RUN

CENTER: ( 3.5          , 1.5          )  RADIUS:   5.70088
CENTER: ( 2     ,-1     )  RADIUS:   4.47214
CENTER: ( .6           , .5           )  RADIUS:   6.57343
POINTS ARE COLLINEAR

OUT OF DATA  IN LINE 60
```

LINEAR SYSTEMS

Exercise 8 — The Interative Method

This exercise is an introduction to an iterative process that can be applied to non-linear systems as well as to linear systems. Students should be encouraged to investigate the method further by studying the suggested references.

The flowchart in the Student Book is sufficient for writing the following program. We have used the DEF FN (defining a function) statement in this program. If your students have not used this statement, this will give them a good introduction to its use.

Example Program
Exercise 8.

```
10   REM--A PROGRAM TO FIND THE SOLUTION OF A SECOND ORDER LINEAR
11   REM--SYSTEM BY AN ITERATIVE PROCESS.  ON THE DATA LINE ENTER
12   REM--THE DESIRED SOLUTION TOLERANCE T, AND THE COEFFICIENTS OF
13   REM--THE GENERAL FORM OF THE SYSTEM.
14   DIM X[200],Y[200]
20   DATA .0001,39,1,-6,25,22,-13,.0001,2.135,86.93,.7394,.8113,4.793
21   DATA -6.843
30   READ T,A1,B1,C1,A2,B2,C2
35   M1=-A1/B1
40   M2=-A2/B2
50   DEF FNA(X)=(C1-A1*X)/B1
60   DEF FNB(Y)=(-C1-B1*Y)/A1
70   DEF FNC(X)=(-C2-A2*X)/B2
80   DEF FND(Y)=(-C2-B2*Y)/A2
90   I=1
100   X[I]=1
110   IF ABS(M1)>ABS(M2) THEN 180
120   Y[I]=FNA(X[I])
130   I=I+1
140   X[I]=FND(Y[I-1])
150   Y[I]=FNA(X[I])
160   IF ABS(X[I]-X[I-1])<T OR ABS(Y[I]-Y[I-1])<T THEN 230
170   GOTO 130
180   Y[I]=FNC(X[I])
190   I=I+1
200   X[I]=FNB(Y[I-1])
210   Y[I]=FNC(X[I])
220   IF ABS(X[I]-X[I-1])<T OR ABS(Y[I]-Y[I-1])<T THEN 230
225   GOTO 190
230   PRINT "(";X[I];",";Y[I];")"
235   GOTO 30
240   END

RUN

(  .142858      ,  .428571      )
(  9.8068       ,-.232349       )

OUT OF DATA   IN LINE 30
```

Exercise 9 — Applying the Gaussian Algorithm

The discussion of the Gaussian Algorithm in the Student Book is detailed and thorough as far as it goes. No attempt is made to deal with round-off error and other problems which might be encountered. Encourage your students to study the suggested reference for a more extensive treatment of this topic.

Micro Flowchart
Exercise 9(a).





Example Program
Exercise 9(a).

```
10   REM--A PROGRAM TO SOLVE AN NTH ORDER LINEAR SYSTEM OF EQUATIONS
11   REM--USING THE GAUSSIAN ALGORITHM TO FORM A TRIANGULAR COEFFECIENT
12   REM--MATRIX AND BACK SOLVING PROCESS TO FIND THE SOLUTION.
13   REM--ON THE DATA LINE ENTER THE ORDER NUMBER OF THE SYSTEM FOLLOWED
14   REM--BY THE ELEMENTS OF THE COEFFICIENT MATRIX.
50   DATA 2,39,1,6,25,22,13,2,2.135,86.93,-.7394,.8113,4.793
51   DATA 6.843,3,2,-1,3,1,-1,3,-2,0,1,1,1,-2,4,3,7,-2,-1,2,4,-2,1
52   DATA -3,5,-2,2,-5,2,-1,4,-1,3,1,3
80   READ N
90   FOR I=1 TO N
100   FOR J=1 TO N+1
110   READ A[I,J]
120   NEXT J
130   NEXT I
140   I=J=1
150   R=1/A[I,I]
```

```
160  A[I,J]=A[I,J]*R
170  IF J=N+1 THEN 200
180  J=J+1
190  GOTO 160
200  K=I+1
210  J=1
220  IF K=N+1 THEN 360
230  M=-A[K,I]
240  A[K,J]=A[I,J]*M+A[K,J]
250  IF J=N+1 THEN 280
260  J=J+1
270  GOTO 240
280  IF K=N THEN 320
290  K=K+1
300  J=1
310  GOTO 230
320  I=I+1
330  J=I
340  K=I
350  GOTO 150
360  I=N
370  J=N+1
380  X[I]=A[I,J]
390  IF J=I+1 THEN 420
400  J=J-1
410  X[I]=X[I]-A[I,J]*(X[J])
415  GOTO 390
420  PRINT "X(";I;") =";X[I]
430  IF I=1 THEN 455
440  I=I-1
450  GOTO 370
455  PRINT
456  GOTO 80
460  END

RUN

X( 2     ) =  .428571
X( 1     ) =  .142857

X( 2     ) =-.252262
X( 1     ) =  9.92492

X( 3     ) =  14.
X( 2     ) =  3
X( 1     ) =-19.

X( 4     ) =-.208696
X( 3     ) =-.426087
X( 2     ) =-.313044
X( 1     ) =  1.04348

OUT OF DATA  IN LINE 80
```

Macro Flowchart
Exercise 9(b).




1
2
3
4
M ← -Aki
Akj ← Aij(M)+Akj
j ← j+1
NO
j = n+1
YES
NO
k = n
YES
i ← i+1
NO
i = n
YES
i ← 1
OUTPUT
xi, n+1
i ← i+1
NO
i = n
YES
STOP

Example Program
Exercise 9(b).

```
10  REM--A PROGRAM THAT SOLVES A SYSTEM OF EQUATIONS BY THE GAUSS-
20  REM--JORDAN ELIMINATION METHOD.  ON THE DATA LINE, ENTER,N,THE
30  REM-- ORDER OF THE SYSTEM, AND THE COEFFECIENT MATRIX
40  REM--ROW BY ROW.
50  DATA 2,39,1,6,25,22,13,2,2.135,86.93,-.7394,.8113,4.793,6.843
60  DATA 3,2,-1,3,1,-1,3,-2,0,1,1,1,-2,4,3,7,-2,-1,2,4,-2,1,-3,5,-2,2,-8
70  DATA -1,4,-1,3,1,3
80  READ N
90  FOR I=1 TO N
100  FOR J=1 TO N+1
110  READ A[I,J]
120  NEXT J
130  NEXT I
140  I=1
150  J=1
160  K=1
170  R=1/A[I,I]
180  A[I,J]=A[I,J]*R
190  IF J=N+1 THEN 220
200  J=J+1
210  GOTO 180
220  IF K=I THEN 310
230  IF K=N+1 THEN 360
240  J=1
250  M=-A[K,I]
260  A[K,J]=A[I,J]*M+A[K,J]
270  IF J=N+1 THEN 300
280  J=J+1
290  GOTO 260
300  IF K=N THEN 330
310  K=K+1
320  GOTO 220
330  IF I=N THEN 360
340  I=I+1
350  GOTO 150
360  FOR I=1 TO N
370  PRINT "X(";I;")=";A[I,N+1]
380  NEXT I
385  GOTO 80
390  END

RUN

X( 1 )= .142857
X( 2 )= .428571
X( 1 )= 9.92492
X( 2 )=-.252262
X( 1 )=-19.
X( 2 )= 3
X( 3 )= 14.
X( 1 )= 1.04348
X( 2 )=-.313044
X( 3 )=-.426087
X( 4 )=-.208696

OUT OF DATA  IN LINE 80
```

Exercise 10 — Solving Linear Systems by Use of Matrices

The following program can be written directly from the flowchart in the Lab Book.

Example Program
Exercise 10.

```
10  REM--A PROGRAM TO SOLVE LINEAR SYSTEMS USING MATRICES.
20  REM--ON THE DATA LINE ENTER THE ORDER OF THE SYSTEM,
30  REM--FOLLOWED BY THE COEFFICIENTS OF THE SYSTEM.
40  DATA 2,39,1,25,22,6,13
50  DATA 2,2.1356,86.93,.8113,4.793,-.7394,6.843
60  DATA 3,2,-1,3,-1,3,-2,1,1,1,1,0,-2,4,3,7,-3,-1,4,-2,1,-3
70  DATA -2,2,-5,2,4,-1,3,1,2,5,-1,3
80  READ N
90  MAT  READ A[N,N],K[N,1]
100  MAT S=ZER[N,1]
110  MAT I=ZER[N,N]
120  MAT I=INV(A)
130  MAT S=I*K
140  PRINT "THE SOLUTION TO THE SYSTEM IS:"
150  MAT  PRINT S
160  GOTO 80
170  END


RUN

THE SOLUTION TO THE SYSTEM IS:

 .142857

 .428571
THE SOLUTION TO THE SYSTEM IS:

 9.9254

-.252342
```

```
THE SOLUTION TO THE SYSTEM IS:

-19.
 3
 14
THE SOLUTION TO THE SYSTEM IS:

 1.03209
-.370767
-.43672
-.188948

OUT OF DATA  IN LINE 80
```

LINEAR PROGRAMMING

Exercise 11 — A Manufacturing Problem

Micro Flowchart
Exercise 11.



NOTE THAT WE HAVE STATED THE NEGATION OF EACH RESTRAINT. THIS WAS DONE TO BE ABLE TO LIST THE CONDITIONAL "GO TO" STATEMENTS IN THE PROGRAM IN SUCCESSION.


1
2
3
4
5
I ← I+1
B←B+1
NO
B = 26
YES
A←A+1
NO
A = 18
YES
OUTPUT
X, Y, P
STOP

Example Program
Exercise 11.

```
10  REM--A PROGRAM TO SOLVE A LINEAR , TWO VARIABLE , PROGRAMMING
11  REM--PROBLEM.  ON THE DATA LINE ENTER THE UPPER BOUNDS, A AND B,
12  REM--OF THE TWO VARIABLES.  ON LINE 60 ENTER THE
13  REM--CONSTRAINTS ON THE VARIABLES.
14  DIM P[100]
40  P=A=0
50  I=1
55  B=0
60  IF 5*A+15*B<120 THEN 130
70  IF 70*A+50*B>1300 THEN 200
80  IF 8*A+20*B>300 THEN 200
90  IF 20*A+20*B>400 THEN 200
100  P[I]=150*A+250*B
110  IF P[I]>P THEN 160
120  I=I+1
130  IF B=26 THEN 200
140  B=B+1
150  GOTO 60
160  P=P[I]
170  X=A
180  Y=B
190  GOTO 120
200  IF A=18 THEN 230
210  A=A+1
220  GOTO 55
230  PRINT "THE MAXIMUM PROFIT ";P;"IS OBTAINED FOR MANUFACTURING"
240  PRINT X;"LOTS OF TYPE A TOYS AND";Y;"LOTS OF TYPE B TOYS."
250  END
```

```
RUN

THE MAXIMUM PROFIT  4100    IS OBTAINED FOR MANUFACTURING
 9    LOTS OF TYPE A TOYS AND 11   LOTS OF TYPE B TOYS.

DONE
```

NOTES

NOTES

NOTES